
*********************************************** CONTENIDO DE LA LIBRERÍA 
************************************************

* Esta librería contiene 21 componentes, de los cuales 18 se pueden simular.

******  RESISTENCIAS
* R1/8W
* R1/4W
* Rajustable de precisión

****** CONDENSADORES 
* Condensador cerámico de 10nF tipo lenteja
* Condensador cerámico miniatura de 100nF
* Condensador cerámico de 22nF
* Condensador electrolítico de 10uF
* Condensador electrolítico de 100uF

****** DIODOS
* Diodo de pequeña señal 1N4148
* Diodo rectificador de 1A 1N4007
* Diodo zéner de 5V1 y 250mW
* Puente rectificador modelo B80 C1000 de 50V de tensión directa y 0.8Amperios

****** TRANSISTORES
* Transistor de pequeña potencia BC547B (NPN)
* Transistor de mediana potencia 2N2219 (NPN)

****** CIRCUITOS INTEGRADOS
* Regulador integrado LM78L05 de +5V y 100mA
* Doble temporizador NE556

****** VARIOS
* Fusible (se simula sin parámetros)
* Transformador modelo Crovisa 22A15 de 15V-0-15V, de una potencia de 2.8VA y 0.2Amp

****** CONECTORES (sin modelo,no se simulan)
* Conector de dos terminales tipo poste
* Conector de dos terminales tipo regleta
* Conector de dos terminales tipo tornillo

*****************************************************************************************************************************
**********
************************************************* COMIENZO DE LA LIBRERIA 
**************************************************
*****************************************************************************************************************************
**********

**************************************************** RESISTENCIAS 
************************************************************

*** RESISTENCIA FIJA DE 1/8 DE VATIO
* Este componente no tiene ningúna propiedad definida hasta ahora, por tanto no tiene modelo.
* La definición se encuentra dentro de la librería de símbolos "ANALOEVA.OLB"
* PspiceTemplate = R^@REFDES %1 %2 @VALUE
* Referencia = R Valor = 1K
* Footprint = R1/8WTIPO1

*** RESISTENCIA FIJA DE 1/4 DE VATIO
* Este componente no tiene ningúna propiedad definida hasta ahora, por tanto no tiene modelo.
* La definición se encuentra dentro de la librería de símbolos "ANALOEVA.OLB"
* PspiceTemplate =R^@REFDES %1 %2 @VALUE
* Referencia = R Valor = 1K
* Footprint = R1/4W

*** RESISTENCIA AJUSTABLE DE PRECISIÓN
* Este componente no tiene ningúna propiedad definida hasta ahora, por tanto no tiene modelo.
* La definición se encuentra dentro de la librería de símbolos "ANALOEVA.OLB"
* PspiceTemplate = R^@REFDES %1 %2 {@VALUE*@SET+.001}
* Referencia = R Valor = 1K      SET = 0.5 (Valor medio)
* Footprint = RAJUSTPRECISIONTP1

**************************************************** CONDENSADORES 
************************************************************

*** CONDENSADOR CERÁMICO DE LENTEJA DE 10NF
* Este componente no tiene ningúna propiedad definida hasta ahora, por tanto no tiene modelo.
* La definición se encuentra dentro de la librería de símbolos "ANALOEVA.OLB"
* PspiceTemplate = C^@REFDES %1 %2 ?TOLERANCE|C^@REFDES| @VALUE ?IC/IC=@IC/
* Referencia = C Valor = 10nF        IC = 0V
* Footprint = CCERAMICO5LC4SP

*** CONDENSADOR CERÁMICO MINIATURA DE 100NF
* Este componente no tiene ningúna propiedad definida hasta ahora, por tanto no tiene modelo.
* La definición se encuentra dentro de la librería de símbolos "ANALOEVA.OLB"
* PspiceTemplate = C^@REFDES %1 %2 ?TOLERANCE|C^@REFDES| @VALUE ?IC/IC=@IC/
* Referencia = C Valor = 100nF        IC = 0V
* Footprint = CCERAMICO4LC3AN5SP

*** CONDENSADOR CERÁMICO DE 22NF
* Este componente no tiene ningúna propiedad definida hasta ahora, por tanto no tiene modelo.
* La definición se encuentra dentro de la librería de símbolos "ANALOEVA.OLB"
* PspiceTemplate = C^@REFDES %1 %2 ?TOLERANCE|C^@REFDES| @VALUE ?IC/IC=@IC/
* Referencia = C Valor = 22nF        IC = 0V
* Footprint = CCERAMICO6.8LC5SP

*** CONDENSADOR ELECTROLÍTICO DE 10UF
* Este componente no tiene ningúna propiedad definida hasta ahora, por tanto no tiene modelo.
* La definición se encuentra dentro de la librería de símbolos "ANALOEVA.OLB"
* PspiceTemplate = C^@REFDES %1 %2 ?TOLERANCE|C^@REFDES| @VALUE ?IC/IC=@IC/
* Referencia = C Valor = 10uF        IC = 0V
* Footprint = CELECTROLITICO5.2CU4SP

*** CONDENSADOR ELECTROLÍTICO DE 100UF
* Este componente no tiene ningúna propiedad definida hasta ahora, por tanto no tiene modelo.
* La definición se encuentra dentro de la librería de símbolos "ANALOEVA.OLB"
* PspiceTemplate = C^@REFDES %1 %2 ?TOLERANCE|C^@REFDES| @VALUE ?IC/IC=@IC/
* Referencia = C Valor = 100uF        IC = 0V
* Footprint = CELECTROLITICO6.4CU3SP

******************************************************* DIODOS 
**************************************************************

*** DIODO 1N4148
* Diodo de pequeña señal para aplicaciones generales con una corriente máxima (IF) de 20mA
.model D1N4148  D(Is=2.682n N=1.836 Rs=.5664 Ikf=44.17m Xti=3 Eg=1.11 Cjo=4p
+               M=.3333 Vj=.5 Fc=.5 Isr=1.565n Nr=2 Bv=100 Ibv=100u Tt=11.54n)
* PspiceTemplate = D^@REFDES %1 %2 @MODEL ~AREA|?AREA/@AREA/||@AREA|
* Referencia = D Valor = D1N4148 
* Footprint = DIODO1/4W
.ENDS
*$

*** DIODO 1N4007
* Diodo rectificador para corriente máxima  (IF) de 1A
.model D1N4007 D(Is=14.11n N=1.984 Rs=33.89m Ikf=94.81 Xti=3 Eg=1.11
+ Cjo=25.89p M=.44 Vj=.3245 Fc=.5 Bv=1500 Ibv=10u Tt=5.7u) 
* PspiceTemplate = D^@REFDES %1 %2 @MODEL ~AREA|?AREA/@AREA/||@AREA|
* Referencia = D Valor = D1N4007
* Footprint = DIODO1/2W

.ENDS
*$

*** DIODO ZÉNER DE 5.1V
* Diodo zéner de una corriente máxima inversa (IR) de 40mA y una potencia máxima de 250mW
.model D5V1 D(Is=1.004f Rs=.5875 Ikf=0 N=1 Xti=3 Eg=1.11 Cjo=160p M=.5484
+ Vj=.75 Fc=.5 Isr=1.8n Nr=2 Bv=5.1 Ibv=5m Nbv=1.1779
+ Ibvl=1.1646m Nbvl=21.894 Tbv1=176.47u)
* PspiceTemplate = D^@REFDES %1 %2 @MODEL ~AREA|?AREA/@AREA/||@AREA|
* Referencia = D Valor = D5V1
* Footprint = DIODO1/4W
.ENDS
*$

*** PUENTE RECTIFICADOR MODELO B50 C1000
* Puente rectificador de 50V de tensión directa y 1Amperio
.SUBCKT B50_C1000 - + AC ACC 

D_D1         - AC D1NAB 
D_D3         - ACC D1NAB 
D_D4         ACC + D1NAB 
D_D2         AC + D1NAB

.model D1NAB D(Is=1.004f Rs=.5875 Ikf=0 N=1 Xti=3 Eg=1.11 Cjo=160p M=.5484
+ Vj=.75 Fc=.5 Isr=1.8n Nr=2 Bv=50 Ibv=5m Nbv=1.1779
+ Ibvl=1.1646m Nbvl=21.894 Tbv1=176.47u)
* PspiceTemplate = X^@REFDES %- %+ %AC %ACC @MODEL
* Referencia = P Valor = B50 C1000
* Footprint = PUENTE0.8A
.ENDS
*$

***************************************************** TRANSISTORES 
**************************************************************

*** TRANSISTOR NPN BC547B
* Sus principales caracteristicas son: Vce max = 45V
* Ic max = 100mA
* Pot max = 500mW
* Fmax = 100Mhz
* hfe = 200-450
* Ic (hfe) = 2mA

* Modelo del transistor BC547B
.model BC547B   NPN(Is=7.049f Xti=3 Eg=1.11 Vaf=62.79 Bf=374.6 Ise=68f Ne=1.576
+               Ikf=81.57m Nk=.4767 Xtb=1.5 Br=1 Isc=12.4f Nc=1.835 Ikr=3.924
+               Rc=.9747 Cjc=5.25p Mjc=.3147 Vjc=.5697 Fc=.5 Cje=11.5p 
+               Mje=.6715 Vje=.5 Tr=10n Tf=410.2p Itf=1.491 Xtf=40.06 Vtf=10)
* PspiceTemplate = Q^@REFDES %c %b %e @MODEL ?AREA/@AREA/
* Referencia = T Valor = BC547B 
* Footprint = TO92
.ENDS
*$

*** TRANSISTOR NPN 2N2219
* Sus principales caracteristicas son: Vce max = 40V
* Ic max = 600mA
* Pot max = 800mW
* Fmax = 300Mhz
* hfe = 75-300

* Ic (hfe) = 10mA

* Modelo del transistor 2N2219
.model Q2N2219 NPN(Is=14.34f Xti=3 Eg=1.11 Vaf=74.03 Bf=255.9 Ne=1.307
+ Ise=14.34f Ikf=.2847 Xtb=1.5 Br=6.092 Nc=2 Isc=0 Ikr=0 Rc=1
+ Cjc=7.306p Mjc=.3416 Vjc=.75 Fc=.5 Cje=22.01p Mje=.377 Vje=.75
+ Tr=46.91n Tf=411.1p Itf=.6 Vtf=1.7 Xtf=3 Rb=10)
* PspiceTemplate = Q^@REFDES %c %b %e @MODEL
* Referencia = T Valor = Q2N2219
* Footprint = TO5
.ENDS
*$

************************************************** CIRCUITOS INTEGRADOS 
***********************************************************

**************************************** REGULADOR INTEGRADO LM78L05 *************************

* Sus principales caracteristicas son: Vs = +5V
* Is max = 100mA
* Modelo LM78L05
.SUBCKT LM78L05 Input Output Ground
   x1 Input Output Ground x_LM78XX PARAMS:
+     Av_feedback=1665, R1_Value=1020
* PspiceTemplate = X^@REFDES %IN %OUT %GND @MODEL
* Referencia = IC Valor = LM78L05 
* Footprint = LM78LXX
.ENDS
*$

* Estabilizador base de la familia 78xx
.SUBCKT x_LM78XX Input Output Ground PARAMS:
+       Av_feedback=1665, R1_Value=1020
*
* SERIES 3-TERMINAL POSITIVE REGULATOR
Vbg 100 0 DC 7.4V
Sbg (100,101),(Input,Ground) Sbg1
Rbg 101 0 1 TC=1.612E-5,-2.255E-6
Ebg (102,0),(Input,Ground) 1
Rreg 102 101 7k
.MODEL Sbg1 VSWITCH (Ron=1 Roff=1MEG Von=3.7 Voff=3)

Rfb 9 8 1MEG
Cfb 8 Ground 265PF
Eopamp 105 0 VALUE={2250*v(101,0)+Av_feedback*v(Ground,8)}
Vgainf 200 0 {Av_feedback}
Rgainf 200 0 1
*Eopamp 105 0 POLY(3),(101,0),(Ground,8),(200,0) 0 2250 0 0 0 0 0 0 1
Ro 105 106 1k
D1 106 108 Dlim
D2 107 106 Dlim
.MODEL Dlim D (Vj=0.7)
Vl1 102 108 DC 1
Vl2 107 0 DC 1

Q1 Input 5 6 Npn1
Q2 Input 6 7 Npn1 10
.MODEL Npn1 NPN (Bf=50 Is=1E-14)
* Efb Input 4 VALUE={v(Input,Ground)+v(0,106)}

Efb Input 4 POLY(2),(Input,Ground),(0,106) 0 1 1
Rb 4 5 1k TC=0.003
Re 6 7 2k
Rsc 7 9 0.275 TC=1.136E-3,-7.806E-6
Rout 9 Output 0.008
*
* Current Limit
*
Rbcl 7 55 290
Qcl 5 55 9 Npn1
Rcldz 56 55 10k
Dz1 56 Input Dz
.MODEL Dz D (Is=0.05p Rs=3 Bv=7.11 Ibv=0.05u)
.ENDS
*$
*

*********************************************** NEE556 *******************************************
* Se trata de un doble temporizador del tipo NE555
* Sus características son:
* Frecuencia máxima de trabajo = 3Mhz
* Tecnología = Bipolar
* Modelo del NE555
.subckt 555d 1 2 3 4 5 6 7 8
+ params:maxfreq=3e6
  r1 8     5    13k
  r2 5     botm 13k
  r3 botm  0    13k
  m1 7 qb 0 0 nchan l=2u w=1000u
  otop 6     5  cmp dgtlnet=r io_std
  obot botm  2  cmp dgtlnet=s io_std
  ud1 dlyline 8 1 s sd dlymod io_std
  ud2 dlyline 8 1 r rd dlymod io_std
  u1 srff(1) 8 1 strt 4 hi sd rd 3 qb t_srff io_555
  uhigh stim(1,1) 8 1  hi io_stm 0s 1
  ustrt stim(1,1) 8 1  strt io_stm 0s 0 1ns 1
.model nchan nmos cgbo=1p cgdo=1p cgso=1p
.model dlymod udly(dlymn={.5/maxfreq}
+ dlyty={.5/maxfreq}
+ dlymx={0.5/maxfreq})
.model cmp doutput(
+ s0name=0 s0vlo=-500 s0vhi=0
+ s1name=1 s1vlo=   0 s1vhi=500)
.model io_555 uio (
+ drvh=96.4 drvl=104
+ atod1="atod_555" atod2="atod_555"
+ atod3="atod_555" atod4="atod_555"
+ dtoa1="dtoa_555" dtoa2="dtoa_555"
+ dtoa3="dtoa_555" dtoa4="dtoa_555")
.model t_srff ugff (tppcqlhty=120ns)
* PspiceTemplate = 
* X^@REFDES %GND %TRIGGER %OUTPUT %RESET %CONTROL %THRESHOLD %DISCHARGE 
%VCC @MODEL PARAMS: MAXFREQ=@MAXFREQ
* Referencia = IC Valor = NE556
* Footprint = DIP14
.ends
*$
.subckt atod_555  a d  dpwr dgnd
+ params: capacitance=0
*
  o0  a dgnd do555 dgtlnet=d io_std

  c1  a 0 {capacitance+0.1pf}
.ends
.subckt dtoa_555  d a  dpwr dgnd
+ params: drvl=0 drvh=0 capacitance=0
*
  n1  a dgnd dpwr din555 dgtlnet=d io_std
  c1  a 0 {capacitance+.1pf}
.ends
*$
.model din555 dinput (
+ s0name="0" s0tsw=0.7ns s0rlo=100 s0rhi=1meg
+ s1name="1" s1tsw=0.7ns s1rlo=1meg s1rhi=300
+ s2name="x" s2tsw=0.7ns s2rlo=200 s2rhi=200
+ s3name="r" s3tsw=0.7ns s3rlo=200 s3rhi=200
+ s4name="f" s4tsw=0.7ns s4rlo=200 s4rhi=200
+ s5name="z" s5tsw=0.7ns s5rlo=200k s5rhi=200k
+ )
*$
.model DO555 doutput (
+ s0name="X" s0vlo=0.8 s0vhi=2.0
+ s1name="0" s1vlo=-1.5s1vhi=0.8
+ s2name="R" s2vlo=0.8 s2vhi=1.4
+ s3name="R" s3vlo=1.3 s3vhi=2.0
+ s4name="X" s4vlo=0.8 s4vhi=2.0
+ s5name="1" s5vlo=2.0 s5vhi=50.0
+ s6name="F" s6vlo=1.3 s6vhi=2.0
+ s7name="F" s7vlo=0.8 s7vhi=1.4
+ )
*$

****************************************************************

****************************************************** VARIOS ***********************************************************

**************************************************** FUSIBLE ****************************************
* Solo se trata del símbolo que se puede simular con una resistencia
* del hilo de 1 miliohmio. No se simula la corriente máxima ni cuando
* se funde por el calor.
* Características: Rhilo = 1mili ohmio

* Subcircuito de la resistencia .
.SUBCKT FUS 1 2  PARAMS: Rp1 = 1m
R_R1         1 3 {Rp1} TC1=0.0033333333
R_R2         3 2 1p TC1=0.0033333333
* PspiceTemplate = X^@REFDES %1 %2 @MODEL PARAMS: ?RP1|Rp1=@RP1||Rp1=1m|
* Referencia = T Valor = Fusible
* Footprint = FUSIBLE
.ENDS FUS
*$

****************************************** TRANSFORMADOR DE 15V Y 0.2AMP ************************************
* Transformador de la casa Crovisa, modelo CROVISA 22A15
* Sus características son: Tensión del primario = 220V
* Tensiónes del secundario = 15-0-15
* Potencia máxima = 2.8VA
* Corriente máxima efícaz (15-0) = 180mA
* Resistencia del primario = 1666 ohmios
* Resistencia secundario (15V-15V) = 91 ohmios

.SUBCKT Crovisa22A15 220V 0P 15V  0S 15Vv 
R_RPN1         5 2 251K 
R_RHP2         220V 2 833 
R_RHS1         6 15Vv 45.5
Kn_K1         L_L1  L_L3 L_L4   1 KTRAFO22A15
R_RHP1         0P 5 833
R_Rsi1         0P 15Vv 10Meg 
R_Rsi2         0P 0S 10Meg 
R_Rsi3         0P 15V 10Meg 
R_RHS2         3 15V 45.5
L_L1         2 5  3947  
L_L3         3 0S  355
L_L4         0S 6  355
* PspiceTemplate = X^@REFDES %220V %0P %15V %0S %15Vv @MODEL
* Referencia = T Valor = Trafo crovisa 22A15
* Footprint = TRAFO15V-0.2A
.ENDS    Crovisa22A15

* Nucleo cuadrado de hierro templado
* Modelo del nucleo del trafo Crovisa 22A15
.MODEL KTRAFO22A15 CORE
+ MS=1.3713E6
+ A=93.002
+ C=72.674E-3
+ K=17.784
+ AREA=1.695
+ PATH=13.37
*$

* By Tino Fernández
* Spain 2009

*******************************************************************************************
***************************************** FIN *********************************************
*******************************************************************************************

